How to convert a system of parametric equations?

I have a system of parametric equations (t is a parameter)

    z = ((1 - t)^2 (1 + 4 t - 2 t^2))/(2 (1 - 2 t + 2 t^2))
y = ((-1 + t) Sqrt[1 - t^2] (1 - 3 t + t^2))/(1 - 2 t + 2 t^2)
{t, 0.7071, 1}


I am going to find the normal equation of z=f(y) without parameter t. How can I do this in Mathematica? As I told before, I found only y=f(z) as analytical expression like:

    Y = ((Sqrt[3] a1 - 6 a2) Sqrt[-a1^2 + 4 Sqrt[3] a1 (2 + a2) -
12 a2 (4 + a2)] (-a1^2 + 4 Sqrt[3] a1 (-1 + a2) -
12 (-4 - 2 a2 + a2^2)))/(
96 Sqrt[3] (a1^2 - 4 Sqrt[3] a1 (1 + a2) + 12 (2 + 2 a2 + a2^2))),


where :

    a1 = Sqrt[(2^(2/3) a^2 - 4 a (-3 + 4 Z) + 2 2^(1/3) (3 + 4 Z)^2)/a],
a2 = Sqrt[
2 - a/(6 2^(1/3)) - (8 Z)/3 + (8 Sqrt[3] Z)/a1 - (3 + 4 Z)^2/(
3 2^(2/3) a)],
a = (-54 + Sqrt[k] + 1080 Z - 576 Z^2 + 128 Z^3)^(1/3),
k = -139968 Z + 1150848 Z^2 - 1396224 Z^3 + 470016 Z^4 -221184 Z^5


but for my task I need z=f(y).

• Have you looked at Eliminate[]? May 19, 2015 at 15:43
• yes, but there is no elimination. May 19, 2015 at 15:46
• What do you mean? Is there no result returned? May 19, 2015 at 15:49
• Welcome to Mathematica.SE! I suggest that: 1) You take the introductory Tour now! 2) When you see good questions and answers, vote them up by clicking the gray triangles, because the credibility of the system is based on the reputation gained by users sharing their knowledge. Also, please remember to accept the answer, if any, that solves your problem, by clicking the checkmark sign! 3) As you receive help, try to give it too, by answering questions in your area of expertise. May 19, 2015 at 15:49
• Please show your code in the Question. May 19, 2015 at 15:50

Here is the implicit equation satisfied by (y,z).

ratpolys = {z - (1 - t)^2 (-2 t^2 + 4 t +
1)/(2 (2 t^2 - 2 t + 1)),
y - (t - 1) *Sqrt[1 - t^2]*(t^2 - 3 t + 1)/(2 t^2 - 2 t + 1)};
gb = First[
GroebnerBasis[ratpolys, {y, z}, t,
MonomialOrder -> EliminationOrder]]

(* Out[93]= -135 y^4 - 1728 y^6 + 1024 y^8 - 720 y^4 z + 1536 y^6 z -
360 y^2 z^2 - 4768 y^4 z^2 + 4096 y^6 z^2 + 160 z^3 - 112 y^2 z^3 +
3840 y^4 z^3 + 720 z^4 - 6176 y^2 z^4 + 6400 y^4 z^4 - 320 z^5 +
3584 y^2 z^5 - 2880 z^6 + 4608 y^2 z^6 + 1280 z^7 + 1280 z^8 *)


One can use Solve to get branches of algebraic roots for either variable in terms of the other.

• Ah, this was what I was going to suggest to the OP after verifying that Eliminate[] did not work for his example… May 20, 2015 at 0:42
• ...but Solve cant get algebraic roots z=f(y): Solve[-135 y^4 - 1728 y^6 + 1024 y^8 - 720 y^4 z + 1536 y^6 z - 360 y^2 z^2 - 4768 y^4 z^2 + 4096 y^6 z^2 + 160 z^3 - 112 y^2 z^3 + 3840 y^4 z^3 + 720 z^4 - 6176 y^2 z^4 + 6400 y^4 z^4 - 320 z^5 + 3584 y^2 z^5 - 2880 z^6 + 4608 y^2 z^6 + 1280 z^7 + 1280 z^8 == 0, z]. Jun 16, 2015 at 7:58
• The result, in terms of Root objects, is comprised of algebraic roots. That's by definition of what an algebraic root is. They are not in terms of explicit radicals but that's a different matter. Jun 17, 2015 at 18:04